U.S. DEPARTMENT OF THE INTERIOR
U.S. GEOLOGICAL SURVEY

HYDRO - A BASIC program and user's guide to model a recorded hydrograph of a debris flow using two different solutions for kinematic wave theory

by

M. Arattano[1], S.H. Cannon[2] and P. S. Powers[2]

Open-File Report 92-358

[1] Work performed under memo of understanding with the U.S. Geological Survey.
Current address is
CNR IRPI Str. Delle Cacce, 73
10135 Torino ITALY

[2] Golden, Colorado

1994

CONTENTS

TABLES

ABSTRACT

We present a BASIC program that computes the parameters of a kinematic wave model for debris-flow motion (Arattano and Savage, 1992) using a recorded hydrograph of a debris flow and the channel cross section measured at the location of the gaging station. The program generates theoretical hydrographs based on two different solutions for kinematic wave theory at the location of the station. These theoretical solutions can then be compared with the recorded hydrograph.

INTRODUCTION

The BASIC computer program HYDRO computes the parameters of the kinematic wave model for debris flow proposed by Arattano and Savage (1992) using one recorded hydrograph from a gaging station along a channel and the value of k calculated from the channel cross section at the gaging station. The solution for kinematic wave theory in Arattano and Savage (1992) was obtained using the method of characteristics. The option also exists to compute the model parameters using a solution for kinematic wave theory proposed by Whitham (1979).

In order to compute the model parameters H and L, a theoretical time of inception of the debris flow is first calculated. H is the maximum height of the initial debris mass, and L is the length of the initial debris mass. H and L are then used to calculate x, the distance between the theoretical point of inception and the gaging station.

Once the model parameters have been calculated, a theoretical hydrograph at the location of the gaging station can be generated by the program. Theoretical hydrographs can be generated using either the solution obtained from the method of characteristics (Arattano and Savage, 1992) or Whitham's (1979) solution.

The program also allows for the comparison of the recorded hydrograph with the hydrographs predicted by the model using either solution. The comparison is done by adding the theoretical time of inception to the recorded time data of the actual hydrograph.

The output of HYDRO consists of files containing Time/Flow-Height pairs. A graphics program, such as GRAPHER™, may be used to draw the hydrographs.

HYDRO.EXE will run on any IBM PC or PC compatible system running DOS 3.0 or above and with 640 kilobytes of memory. HYDRO was written using Microsoft BASIC Professional Development System version 7.1.

SOME USEFUL ADVICE BEFORE RUNNING THE PROGRAM

A previous ordering of the input and output data can facilitate using HYDRO. In particular, it is helpful to decide the names of the output files before running the program to distinguish the type of solution used.

In table 1 we have shown a possible set of file names needed to run the program. We suggest creating a table similar to that shown as table 1 before starting HYDRO and referring to it while running the program. The first column contains a suggested input filename for the Time/Flow-Height (TH) data pairs recorded at the gaging station. The second column contains a suggested output file name for the Corrected Time/Flow-Height (CTH) data pairs (The extension .DAT is needed to use the graphing program GRAPHER™). These are the recorded hydrograph data that will be corrected by adding the theoretical time of inception to the time data, to allow the comparison with the theoretical hydrograph. The third and fourth columns contain suggested file names for Theoretical Time/Flow-Height data for the Method of Characteristics solution (TTHMC) and Whitham's (1979) Solution (TTHWS). These are the files that will be used to draw the theoretical hydrographs.

Together with the executable program, we have provided a file which contains the hydrograph data recorded with time-lapse photography during the descending limb of the largest surge of a debris flow that occurred on June 5, 1983, at a site in Lower Rudd Canyon, Farmington, Utah (Pierson, 1985). This file is named TH, and can be used, together with the data given in table 2, to demonstrate how the program runs. An example of a Time/Height-data file is shown below. The column on the left contains time data, the column on the right contains height data. The first point is the peak flow value and its time is thus assumed equal to zero.

```
0,   2.26
5,   1.95
10,  1.69
17,  1.44
27,  1.13
35,  0.97
```

CONVENTIONS USED

To set apart the information that is entered from the keyboard from the text of this report, the keyboard input is in ***bold italics.*** Single keystroke input like [ENTER] or [ESC] is capitalized and enclosed in brackets. Screen messages are boxed in with double lines and are shaded to appear like a computer display.

INPUT	OUTPUT		
File containing recorded data	File containing corrected recorded data	Files containing theoretical hydrograph data	
		Method of Characteristics	Whitham's Solution
TH	CTH.DAT	TTHMC.DAT	TTHWS.DAT

Table 1 - Example of a set of file names needed to run the program.

s	k	C [m^{1-k}/s]	h_p [m]
0.10	0.46	5.8	2.26

s slope of the channel
k accounts for changes in R with flow height
C roughness coefficient
h_p peak flow height

Table 2 - Field data (s and h_p) and calculated parameters (k, and C) from the largest surge of a debris flow that occurred on June 5, 1983 in lower Rudd Canyon, Farmington, Utah (Pierson, 1985).

```
This is a sample of a screen message or prompt
```

If a problem is encountered while running the program, press and hold the [CTRL] key and the [BREAK] key to exit. This aborts the program and returns the system to DOS.

THE FIRST THING TO DO

Insert the program/data disk into the computer drive. The A: drive is usually a 5.25" drive, and the B: drive a 3.5" drive. To run the program from a floppy drive, change the working drive to the floppy drive by typing the appropriate drive letter followed by a colon (i.e. A:). If you wish to run the program from the hard drive, copy the floppy disk files to the hard disk with the DOS copy command (i.e. copy A:*.* C:*.*).

GETTING STARTED

Start the program by typing ***HYDRO*** and pressing the [ENTER] key. The screen is first cleared and the following title page is displayed:

```
                U.S. DEPARTMENT OF INTERIOR
                  U.S. GEOLOGICAL SURVEY

              HYDRO -- A BASIC program to model
            a recorded hydrograph of a debris flow
                using two different solutions
                  for kinematic wave theory

                             by

          M. Arattano, S.H. Cannon and P.S. Powers


press any key to continue.
```

Press ***any key*** to proceed to the next screen.

The next screen informs the user how to quit the program in case of a problem:

```
* * * IMPORTANT * * *

If you need to quit the program at any time press and hold
  [CTRL] and then press [BREAK].


  to continue with the program

  PRESS ANY KEY
```

press *any key* to proceed.

DATA INPUT

The input to the program consist of, in order, the slope of the channel, s, the parameter k, which accounts for change in hydraulic radius with flow height, the roughness coefficient C, the peak flow height recorded at the gaging station, h_p, and the name of the input file containing the TH pairs taken from the recorded hydrograph (for an example, see table 1, column 1). Figure 1 shows an example of a TH file from a recorded hydrograph.

At this point the user is asked for the name of the file to which the corrected hydrograph data will be written (for an example, see table 1, column 2). The hydrograph data will be corrected by adding the time, t_c, from the theoretical moment of inception of the debris flow to its recording at the gaging station.

```
  Enter the output file name for the corrected hydrograph
  data.
```

The program will now perform the calculation to find the values for t_c, H, L, and x using the solution obtained from the method of characteristics (Arattano and Savage, 1992). Recall that H is the maximum height of the initial debris mass, L is the length of the initial debris mass, and x is the distance between the theoretical point of inception and the gaging station. (See Arattano and Savage (1992) for a detailed explanation of these variables). These values will be shown on the screen when the calculations are complete, as below.

```
corr. coeff. (r) = .9997
Intercept =  9.892

Peak Height
  at the
 Station     U        x       H       L       t      k      C
   (m)     (m/s)     (m)     (m)     (m)     (s)
---------------------------------------------------------------
  2.26     4.44    308.21   6.77   67.66   73.3   0.46   5.8

Enter (Y)es to send the screen to the printer (LPT1), else
PRESS ANY KEY.
```

If the printer is attached but not on line, the program cannot finish execution.

The next prompt will request the number of points used to generate the theoretical hydrograph at the gaging station.

```
Enter the number of points on the theoretical hydrograph.
```

To generate the theoretical hydrograph the program will calculate a user-designated number of Time/Flow-Height pairs. (50 points are usually sufficient).

The calculated points will be stored into a file that can then be read by a graphics program like GRAPHER™. The following screen asks for the name of this file (for an example see column 3 in table 1). (If you use GRAPHER™ to draw the hydrograph, the extension of the filename must be .DAT).

```
Enter the output file name for the theoretical hydrograph
calculated using the method of characteristics.
```

The program now allows the option to compare the result obtained by solving the equations using the method of characteristics with the results obtained using Whitham's (1979) solution. The program now prompts:

```
Enter Y(es) to compare the result with that obtained from
Whitham's solution. Else enter N(o).
```

If this option is selected the program will perform the calculation to find the values for H, L, and x, and to generate a theoretical hydrograph at the gaging station, using Whitham's (1979) solution. The program will request the number of points to be used in generating the theoretical hydrograph and the output filename for the hydrograph data (for an example, see table 1, column 4). When the calculations are complete, the values for H, L, x and other variables will be shown on the screen:

```
corr. coeff. (r) = .9997
Intercept =  9.892

Peak Height
  at the
 Station    U       x        H      L      t      k     C
   (m)    (m/s)    (m)      (m)    (m)    (s)
-----------------------------------------------------------
  2.26    4.30   285.61   6.38   63.78   73.3   0.46   5.8

Enter (Y)es to send the screen to the printer (LPT1), else
press ANY KEY.
```

The next prompt is :

```
Press the SPACE BAR to RUN the program again, or any other
key to EXIT.
```

Pressing any key other than the space bar will exit the program and the following message will be displayed:

```
                         ALL DONE
```

REFERENCES

Arattano, M. and Savage, W.Z., 1992, Kinematic wave theory for debris flows: U.S. Geological Survey Open-File Report 92-290, 39 p.

Pierson, T. C., 1985, Effects of slurry composition on debris flow dynamics, Rudd Canyon, Utah, in Bowles, D.S. ed., Delineation of landslide, flash flood, and debris-flow hazard in Utah: Logan, Utah Water Research Laboratory, Utah State University, General Series UWRL/G-85/03, p. 132-152.

Whitham, G.B., 1979, Lectures on wave propagation: New York, Springer-Verlag, 151 p.

APPENDIX: PROGRAM LISTING

```
DECLARE SUB FileExists (filename$, filenumber%)
DECLARE SUB Calculate1 ()
DECLARE SUB CFiles ()
DECLARE SUB Results1 ()
DECLARE SUB Thydro ()
DECLARE SUB PScreen ()
DECLARE SUB PNote ()
DECLARE SUB Openfiles (hydrographTimes!(), hydrographHeights!(), theight$)
DECLARE SUB Readfile (ct%, n, hydrographTimes!(), hydrographHeights!(), theight$)
DECLARE SUB Hydro (hs!, x!, k!, L!, U!, row%, a!(), H)
DECLARE SUB Hydrow (hs, xx, k, L, U, row%, hydrotheor(), H)

DIM hydrotheor(150, 2)
DIM hydrographTimes(100)
DIM hydrographHeights(100)

CALL PScreen
CALL PNote

CLS

doagn:

INPUT "Enter the slope of the channel. ", s
INPUT "Enter K (exponent of the hydraulic radius). ", k
INPUT "Enter the coefficient C. ", C
INPUT "Enter the peak height of the flow ", h1

CALL Openfiles(hydrographTimes(), hydrographHeights(), theight$)

LOCATE 12, 25
COLOR 31, 0
PRINT "Calculating"
COLOR 7, 0

CALL Calculate1

H = SQR(2 * k * Tdimensional * C * s ^ (3 / 2) * h1 ^ (k + 1) + h1 ^ 2)
hs = h1 / H
xx = (k + 1) / (2 * k * hs) - (1 - k) / (2 * k) * hs
L = H / s
U = C * H ^ k * s ^ .5
```

```
CALL Results1
CALL Thydro
CALL Hydro(hs, xx, k, L, U, row%, hydrotheor(), H)

CLS

retryWhit:

PRINT "Enter (Y)es to compare the result with that obtained from"
INPUT "the Whitham's solution, else enter (N)o. ", z$
z$ = UCASE$(LEFT$(z$, 1))

IF z$ <> "N" AND z$ <> "Y" THEN
      CLS
      PRINT "Please Respond with (Y)es or (N)o."
      GOTO retryWhit:
END IF

IF z$ = "N" THEN GOTO ending:

CLS

H = SQR(2 * k * Tdimensional * C * s ^ (3 / 2) * h1 ^ (k + 1))
hs = h1 / H
xx = (k + 1) / (2 * k * hs)
L = H / s
U = C * H ^ k * s ^ .5

CALL Results1
CALL Thydro
CALL Hydrow(hs, xx, k, L, U, row%, hydrotheor(), H)

ending:

CLS
PRINT "Press the SPACE BAR to RUN program again, or any other key to EXIT."
PRINT "";
s$ = ""

DO WHILE s$ = ""
   s$ = INKEY$
LOOP
```

```
'
' ASCII character  32 is the space  bar.
'
IF s$ = CHR$(32) THEN
  CLS
  CLOSE #1, #2, #3, #4, #5, #6, #7, #8, #9, #10
  datafilename$  = ""
  GOTO doagn:
ELSE
  CLS
  GOTO done1:
END IF

done1:
CALL CFiles
END

'===========================================================================
'                    Do Calculations
'===========================================================================
SUB Calculate1
DIM lht(100), lhh(100)
SHARED ct%, n, hydrographTimes(),  hydrographHeights(),  slopem, intercept, corrcoef
SHARED theight$, k, Tdimensional
'
'Tdimensional has only to be a positive number for the initial value.
'
Tdimensional = 10

loopagain1:
ct% = 0
CALL Readfile(ct%, n, hydrographTimes(),  hydrographHeights(),  theight$)
sumxy = 0
sumx = 0
sumy = 0
sumxi = 0
sumyi = 0
sumx2 = 0
sumy2 = 0
FOR i = 1 TO n
    lht(i) = LOG(hydrographTimes(i))
    lhh(i) = LOG(hydrographHeights(i))
```

```
    '
    'Compute the SUMS necessary to calculate the linear regression line.
    '
    sumxy = sumxy + lht(i) * lhh(i)
    sumx = sumx + lht(i)
    sumy = sumy + lhh(i)
    sumxi = sumxi + lht(i) / ct%
    sumyi = sumyi + lhh(i) / ct%
    sumx2 = sumx2 + lht(i) ^ 2
    sumy2 = sumy2 + lhh(i) ^ 2
NEXT i

'CSCP is the computed sum of the cross product.
cscp = sumxy - (sumx * sumyi)
cssx = sumx2 - (sumx * sumxi)
cssy = sumy2 - (sumy * sumyi)
slopem = cscp / cssx
intercept = sumyi - (slopem * sumxi)
corrcoef = ABS(cscp / (cssx * cssy) ^ .5)
Knew = -1 / (slopem)

IF Knew - k > .005 THEN
  CLOSE #4
  Tdimensional = Tdimensional + .1
  GOTO loopagain1:
ELSE
  GOTO checktwo:
END IF

checktwo:

IF Knew - k < -.005 THEN
  CLOSE #4
  Tdimensional = Tdimensional - .1
  GOTO loopagain1:
ELSE
  GOTO loopout:
END IF

loopout:

FOR i = 1 TO n
    PRINT #5, USING "#####.###     ####.###"; hydrographTimes(i);
hydrographHeights(i)
```

```
NEXT i

END SUB

'=====================================================
=================================
'                    CLOSE FILES
'=====================================================
=================================
SUB CFiles

CLS
LOCATE 11, 17
COLOR 0, 7
PRINT "                              "
LOCATE 12, 17
PRINT "           ALL DONE.           "
LOCATE 13, 17
PRINT "                              "
COLOR 7, 0
BEEP
terminate:
CLOSE #1, #2, #3, #4, #5, #6, #7, #8, #9, #10

END SUB

'=====================================================
==================================
'                    File Exists
'=====================================================
==================================
'
'File already existed, make selection to overwrite, append or select new name.
'
SUB FileExists (filename$, filenumber%)

SHARED cont$, ans$

IF UCASE$(LEFT$(cont$, 1)) <> "Y"THEN
  COLOR 0, 7
  CLOSE filenumber%
  PRINT filename$; " already exits."
  COLOR 7, 0
  ans$ = ""
```

```
    DO WHILE ans$ <> "Y" AND ans$ <> "A" AND ans$ <> "N"
        PRINT "Enter Y = Overwrite, A = Append, or else N to enter a different name. "
        kpress$ = ""

        DO WHILE kpress$ = ""
          kpress$ = INKEY$
        LOOP

        ans$ = UCASE$(kpress$)
    LOOP

    IF ans$ = "N" THEN GOTO done:

    IF ans$ = "Y" THEN
          KILL filename$
          GOTO proceed:
    END IF

    IF ans$ = "A" THEN
          append$ = "A"
          GOTO proceed:
    END IF

    filename$ = ""
    GOTO done:
END IF

proceed:

IF append$ = "A" OR cont$ = "Y" THEN
      CLOSE filenumber%
      OPEN filename$ FOR APPEND AS filenumber%
      append$ = ""
ELSE
      OPEN filename$ FOR OUTPUT AS filenumber%
END IF

done:

END SUB

'==========================================
============================
'                          Hydro
```

```
'=========================================================================
SUB Hydro (hs, x, k, L, U, row%, a(), H) STATIC

' subroutine to compute heights for
' different times at the same x

SHARED ans$

p = hs / (row% + 2)
CLS

agn:

PRINT "Enter the output filename for the theoretical hydrograph calculated"
INPUT "using the method of characteristics. ", filehydro1$

filenumber7% = 7

ON LOCAL ERROR GOTO erHydro:
OPEN filehydro1$ FOR INPUT AS filenumber7%
ON LOCAL ERROR GOTO 0
CLOSE filenumber7%
CALL FileExists(filehydro1$, filenumber7%)
IF ans$ = "N" THEN GOTO agn:

donehydro:

FOR i = 1 TO row%
   a(i, 2) = (hs - (row% - i) * p)
   a(i, 1) = (x - a(i, 2)) / ((k + 1) * (a(i, 2) ^ k))
   PRINT #7, USING "#####.###      ####.###";    a(i, 1) * L / U; a(i, 2) * H
NEXT i

CLOSE #7

EXIT SUB

erHydro:

IF ERR = 53 THEN
      OPEN filehydro1$ FOR OUTPUT AS filenumber7%
      cont$ = "Y"
      RESUME donehydro:
```

```
END IF

IF ERR = 55 THEN
  PRINT "The file name  you  used  is already  open.   Try another  name."
  RESUME agn:
END IF

PRINT "ERROR DETECTED.  "; ERR
PRINT "Correct  the  file problem  and  restart  program."
END

END SUB

'==========================================================
==================================
'                          Hydrow
'==========================================================
==================================
SUB Hydrow (hs, x, k, L, U, row%, a(), H) STATIC

SHARED ans$
p = hs / (row% + 2)

agnhydrow:
filehydro2$ = ""
INPUT "Enter output  filename  for the  theoretical  hydrograph  (dimensional).   ",filehydro2$

ON LOCAL ERROR GOTO erhydrow:
OPEN filehydro2$ FOR INPUT AS #13
ON LOCAL ERROR GOTO 0

CALL FileExists(filehydro2$, 13)
IF ans$  = "N" THEN GOTO agnhydrow:

calchydrow:

FOR i = 1 TO row%
  a(i, 2) = (hs - (row% - i) * p)
  a(i, 1) = x / ((k + 1) * (a(i, 2) ^ k))
  PRINT #13,  USING "#####.###       ####.###";      a(i, 1) * L / U; a(i, 2) * H
NEXT i

CLOSE #13
```

```
EXIT SUB

erhydrow:

IF ERR = 53 THEN
      OPEN filehydro2$ FOR OUTPUT AS 13
      cont$ = "Y"
      RESUME calchydrow:
END IF

IF ERR = 55 THEN
   PRINT "The file name you used is already open. Try another name."
   RESUME agnhydrow:
END IF

PRINT "ERROR DETECTED. "; ERR
PRINT "Correct the file problem and restart program."
END

END SUB

'=============================================
==============================
'                              Open Files
'=============================================
==============================
SUB Openfiles (hydrographTimes(), hydrographHeights(), theight$)
SHARED datafilename$, ct%, Tdimensional, cont%, ans$
ct% = 0
retrytheight:
INPUT "Enter the name of the input file with time, height data. ", theight$

IF theight$ = "?" THEN
   FILES "*.*"
   GOTO retrytheight:
END IF

ON LOCAL ERROR GOTO error2:
OPEN theight$ FOR INPUT AS #4
ON LOCAL ERROR GOTO 0

WHILE NOT EOF(4)
      ct% = ct% + 1
      INPUT #4, hydrographTimes(ct%), hydrographHeights(ct%)
```

```
WEND

CLOSE #4
n = ct%

FOR i = 1 TO n
      hydrographTimes(i)  = hydrographTimes(i)  + Tdimensional
NEXT i

CLS

retry:

IF datafilename$  = ""THEN
  INPUT; "Enter the output filename for the corrected hydrograph data. ",datafilename$
ELSE
  GOTO loopagain:
END IF

CLS
filenumber5%  = 5

ON LOCAL ERROR GOTO eropenfiles:
OPEN datafilename$ FOR INPUT AS filenumber5%
ON LOCAL ERROR GOTO 0
CLOSE filenumber5%

CALL FileExists(datafilename$, filenumber5%)

IF ans$  = "N"THEN
      datafilename$  = ""
      GOTO retry:
END IF

loopagain:

EXIT SUB

eropenfiles:

IF ERR = 53 THEN
      OPEN datafilename$ FOR OUTPUT AS filenumber5%
      RESUME loopagain:
END IF
```

```
EXIT SUB

error2:

IF ERR = 53 THEN
      PRINT "File Not Found.  Try another  filename  or press  [?] for a file list."
      RESUME retrytheight:
END IF

PRINT "ERROR DETECTED.  "; ERR
PRINT "LINE DETECTED.  "; ERL
PRINT "Make sure  of your  file names  and  their  locations,  and  restart."

END SUB

'==================================================
=================================
'                  PRINT SCREEN  NOTES
'==================================================
=================================
SUB PNote
      CLS
      LOCATE 2, 30
      COLOR 15, 0
      PRINT "* * * IMPORTANT * * *"
      LOCATE 10, 10
      COLOR 7, 0
      PRINT "If you  need  to quit the program  at any  time press  and  hold"
      PRINT "[CTRL] and  then  press  [BREAK]. ";
      LOCATE 20, 1
      COLOR 31, 0
      PRINT "To continue  with the  program"
      PRINT
      PRINT "PRESS  ANY KEY"
      COLOR 7, 0

      DO WHILE INKEY$ = ""
      LOOP

END SUB

'
'==================================================
=================================
```

```
'    Print a title on the screen
'=====================================================================
'
SUB PScreen
CLS
PRINT ""
PRINT ""
PRINT "                U.S. DEPARTMENT OF THE INTERIOR"
PRINT "                   U.S. GEOLOGICAL SURVEY"
PRINT ""
PRINT ""
PRINT "             HYDRO -- A BASIC program to model a recorded"
PRINT "                hydrograph of a debris flow using"
PRINT "                   two different solutions"
PRINT "                   for kinematic wave theory"
PRINT ""
PRINT ""
PRINT "                         by"
PRINT ""
PRINT
PRINT "             M. Arattano, S.H. Cannon, and P.S. Powers  "
LOCATE 25, 50
COLOR 31, 0

PRINT "Press any key to continue";
COLOR 7, 0

DO WHILE INKEY$ = ""
LOOP

END SUB

SUB Readfile (ct%, n, hydrographTimes(), hydrographHeights(), theight$)
'=====================================================================
'                  Readfile
'=====================================================================

SHARED Tdimensional

OPEN theight$ FOR INPUT AS #4
```

```
WHILE NOT EOF(4)
  ct% = ct% + 1
  INPUT #4, hydrographTimes(ct%), hydrographHeights(ct%)
WEND

CLOSE #4
n = ct%

FOR i = 1 TO n
  hydrographTimes(i) = hydrographTimes(i) + Tdimensional
  'PRINT hydrographTimes(i)
NEXT i

END SUB

'==========================================================
==================================
'                    PRINT RESULTS TO SCREEN/FILE
'==========================================================
==================================
SUB Results1
SHARED s, k, C, h1, Tdimensional, corrcoef, intercept
SHARED U, xx, L, H
CLS
PRINT USING "Correlation coefficient (r):   #.###        "; corrcoef
PRINT USING "Intercept is:                ####.###        "; intercept
PRINT
PRINT
PRINT "  Peak  "
PRINT " Height   "
PRINT " at the  "
PRINT "Station      U       x       H      L     t      k     C"
PRINT "  (m)      (m/s)     (m)      (m)     (m)    (s) "
PRINT "------------------------------------------------------------------------------"
PRINT USING " ##.##       ##.##      ###.##        ##.##      ##.##    ####.#      #.####
##.##";   h1; U; xx * L; H; L; Tdimensional;  k; C
PRINT
PRINT

retryprinter:

PRINT "Enter (Y)es to send the screen to the printer (LPT1), else PRESS ANY KEY. "
ans$ = ""
```

```
DO WHILE ans$ = ""
  ans$ = INKEY$
LOOP

ans$ = UCASE$(LEFT$(ans$, 1))

IF ans$ = "Y" THEN

   ON LOCAL ERROR GOTO noprinter:
   OPEN "lpt1:" FOR OUTPUT AS #10
   ON LOCAL ERROR GOTO 0

   PRINT #10, USING "Intercept is:              ####.###      "; intercept
   PRINT #10, USING "Correlation coefficient (r):    #.###     "; corrcoef
   PRINT #10, " Peak  "
   PRINT #10, " Height   "
   PRINT #10, " at the  "
   PRINT #10, "Station     U      x      H      L     t      k      C"
   PRINT #10, "  (m)      (m/s)      (m)      (m)      (m)     (s) "
   PRINT #10, "------------------------------------------------------------------------"
   PRINT #10, USING " ##.##       ##.##       ###.##        ##.##       ##.##     ####.#
#.####      ##.##"; h1; U; xx * L; H; L; Tdimensional; k; C
  PRINT #10, CHR$(12)
  CLOSE #10
END IF

EXIT SUB

noprinter:

PRINT "Unable to make the connection to the printer.1448XChgckn printer"
PRINT "and retry."
RESUME retryprinter:

END SUB

'==========================================
==============================
'                    THEORETICAL HYDROGRAPH
'==========================================
==============================
SUB Thydro

SHARED row%
```

```
CLS

inrow:

     INPUT "Enter the number of points on the theoretical hydrograph.   ", row%

      IF row% <= 0 THEN
        PRINT "Number must be greater zero."
        GOTO inrow:
      END IF

      IF row% > 150 THEN
        PRINT "Number must be less than 151 points."
        GOTO inrow:
      END IF

END SUB
```